モーメント法によるダイポールアンテナの解析 (区分正弦ガラーキン法)

MoM Analysis for a Dipole Antenna (Piecewise Sinusoidal Galerkin's Method)

Mathematica 7

平野拓一 (Takuichi Hirano)



General Formula

$$\mathbb{A} = \int_{-\ell/2}^{\ell/2} \frac{\mu\, \mathbb{i}(z')\, e^{-jkr}}{4\pi r} dz' \quad \text{(Vector Potential)}$$

$$G = \frac{e^{-jkr}}{4\pi r} \quad \text{(Green's Function)}$$

$$r = \sqrt{a^2 + (z - z')^2} \quad \text{(Distance between Source Point and Observation Point)}$$

$$\nabla\cdot\mathbb{A} + j\omega\mu\epsilon\,\phi = 0 \quad \text{(Lorentz Condition)}$$

$$\phi = -\frac{\nabla\cdot\mathbb{A}}{j\omega\mu\epsilon}$$

$$\mathbb{E} = -\nabla\phi - j\omega\mathbb{A} = \frac{1}{j\omega\mu\epsilon}\left(\nabla\nabla\cdot\mathbb{A} + k^2\,\mathbb{A}\right)$$

Dipole Antenna

$\mathbf{i} = \hat{z}\, i_z$ より

$$A_z = \int_{-\ell/2}^{\ell/2} \frac{\mu\, i_z(z')\, e^{-jkr}}{4\pi r} dz' = \int_{-\ell/2}^{\ell/2} \mu\, i_z(z')\, G\, dz'$$

$$E_z = \frac{1}{j\omega\mu\epsilon}\left(\frac{\partial^2 A_z}{\partial z^2} + k^2 A_z\right) = \frac{1}{j\omega\epsilon}\int_{-\ell/2}^{\ell/2}\left[\frac{\partial^2 G}{\partial z^2} + k^2 G\right] i_z(z')\, dz'$$

$$E_z + E_z{}^i = 0 \ \ \text{(on PEC)}$$

$$\frac{1}{j\omega\epsilon}\int_{-\ell/2}^{\ell/2}\left[\frac{\partial^2 G(z, z')}{\partial z^2} + k^2 G(z, z')\right] i_z(z')\, dz' + E_z{}^i(z) =$$

0 (Pocklington' s Integralequation)

シンプソンの公式で数値積分する関数を定義

In[1]:=
```
simpi[f_, {ξ_, a_, b_}, n_] := Module[{h = (b - a) / (2 * n), x, i},
   x[i_] := a + h * i;
   h/3 ((f /. {ξ → x[0]}) + 4 Sum[(f /. {ξ → x[2 * i - 1]}), {i, 1, n}] +
       2 Sum[(f /. {ξ → x[2 * i - 2]}), {i, 2, n}] + (f /. {ξ → x[2 * n]})) // N
  ];
```

Parameters of Dipole Antenna

```
l = 0.5; (* アンテナ長 *)
a = 0.001; (* アンテナ半径 *)
volt = 1.; (* 励振電圧 *)
nn = 21; (* 分割数,unknown の数, デルタギャップ給電するから奇数のみ *)
dz = l / (nn + 1);
```

In[2]:=
```
Clear[a];
l = 0.5; (* アンテナ長 *)
sol = Solve[10 == 2 * Log[l/a], a][[1]];
a = a /. sol; (* アンテナ半径 *)
volt = 1.; (* 励振電圧 *)
dz = 0.025;
nn = Floor[l / dz] + Floor[Mod[l / dz + 1, 2]]; (* 分割数,
unknown の数, デルタギャップ給電するから奇数のみ *)
dz = l / (nn + 1);
```

Analysis Start !



In[11]:=

```
Module[{pos, r, r1, r2, r3, zz, vv, z, srcint, k, ii, curdist}, k = 2 π; pos[n_] := dz * n - 1/2;
  r[zo_, zs_] := Sqrt[a^2 + (zo - zs)^2];
  r1[n_, z_] := r[z, pos[n - 1]];
  r2[n_, z_] := r[z, pos[n]];
  r3[n_, z_] := r[z, pos[n + 1]]; srcint[n_, z_] :=
   ((ⅈ 30) (E^(-ⅈ k r2[n,z]) Cos[k (pos[n]-pos[n-1])]/r2[n,z] - E^(-ⅈ k r1[n,z])/r1[n,z]))/Sin[k (pos[n] - pos[n - 1])] +
   ((ⅈ 30) (E^(-ⅈ k r2[n,z]) Cos[k (pos[n+1]-pos[n])]/r2[n,z] - E^(-ⅈ k r3[n,z])/r3[n,z]))/Sin[k (pos[n + 1] - pos[n])];
  zz[m_, n_] := simpi[(Sin[k (z - pos[m - 1])] srcint[n, z])/Sin[k (pos[m] - pos[m - 1])], {z, pos[m - 1], pos[m]}, 10] +
    simpi[(Sin[k (pos[m + 1] - z)] srcint[n, z])/Sin[k (pos[m + 1] - pos[m])], {z, pos[m], pos[m + 1]}, 10];
  vv[i_] := If[i == (nn + 1)/2, -volt, 0];
  Print["******** Now Making Z Matrix... ********"];
  zmat = Table[zz[i, j], {i, 1, nn}, {j, 1, nn}];
  Print["******** Now Making V Matrix... ********"];
  vmat = Table[vv[i], {i, 1, nn}];
  imat = Table[ii[i], {i, 1, nn}];
  Print["******** Now Solving Linear Equations... ********"];
  imat = imat /. Solve[zmat.imat == vmat, imat];
  curdist = Table[{pos[i], imat[[1]][[i]]}, {i, 1, nn}];
  curdist = Prepend[curdist, {-1/2, 0}];
  curdist = Append[curdist, {1/2, 0}];
  Print["Current: ", imat[[1, (nn + 1)/2]]];
  zin = volt/imat[[1, (nn+1)/2]];
  Print["Input Impedance: ", zin];
  Print["Input Admittance: ", 1 / zin];
  Module[{curre, curim}, curre = ListPlot[MapAt[Re, curdist, Table[{i, 2}, {i, 1, nn + 2}]],
      Joined → True, PlotStyle → {Red, AbsoluteThickness[2]}, DisplayFunction → Identity];
   curim = ListPlot[MapAt[Im, curdist, Table[{i, 2}, {i, 1, nn + 2}]], Joined → True,
      PlotStyle → {Green, AbsoluteThickness[2]}, DisplayFunction → Identity];
   Show[{curre, curim}, DisplayFunction → $DisplayFunction]];
  Module[{}, curamp = ListPlot[MapAt[Abs, curdist, Table[{i, 2}, {i, 1, nn + 2}]],
      Joined → True, PlotStyle → {Red, AbsoluteThickness[2]}, Frame → True,
      FrameLabel → {"Position (λ)", "Amplitude (A)"}, DisplayFunction → Identity];
   curpha = ListPlot[Take[MapAt[If[Abs[#1] > 1/10^6, (ArcTan[Re[#1], Im[#1]] 180)/π] &,
        curdist, Table[{i, 2}, {i, 2, nn + 1}]], {2, Length[curdist] - 1}],
      Joined → True, PlotStyle → {Green, AbsoluteThickness[2]}, Frame → True,
      FrameLabel → {"Position (λ)", "Phase (deg)"},
      DisplayFunction → Identity]; Print[Show[{curamp}]];
   Print[Show[{curpha}]];
  ]
]
```

```
******** Now Making Z Matrix... ********

******** Now Making V Matrix... ********

******** Now Solving Linear Equations... ********

Current: 0.00885173 - 0.00414254 i

Input Impedance: 92.6749 + 43.3711 i

Input Admittance: 0.00885173 - 0.00414254 i
```





指向性を求める

In[29]:=

```
cur = Table[{Abs[ imat[[1, i]] ], Arg[ imat[[1, i]] ]}, {i, 1, Length[ imat[[1]] ]}];
η = 120. * π;
k = 2 * π;
pos[n_] := (1 * n)/(nn + 1) - 1/2;
eθ[θ_] := (I * k * η)/(4 * π) * dz * Sin[θ] * Sum[(cur〚i, 1〛 * ⅇ^(ⅈ cur〚i,2〛)) * ⅇ^(ⅈ*k*pos[i+1]*Cos[θ]), {i, 1, Length[cur]}];
(*---- Gain Calculation ----*)
gain[θ_] := Re[4 * π * (eθ[θ]*Conjugate[eθ[θ]])/(2*η)] / Re[volt * Conjugate[imat〚1, (nn+1)/2〛] / 2];
Print["Directivity = ", gain[π/2], " = ", 10 * Log[10, gain[π/2]], " dBi"];
z0 = 50.;
s11 = (zin - z0)/(zin + z0);
Print["Gain (inc. reflection) = ", (1 - Abs[s11]^2) * gain[π/2],
  " = ", 10 * Log[10, (1 - Abs[s11]^2) * gain[π/2]], " dBi"];
(*---- Plot Pattern ----*)
rot[θ_] := {{Cos[θ], -Sin[θ]}, {Sin[θ], Cos[θ]}};
scale[sx_, sy_] := {{sx, 0}, {0, sy}};
eθlist = Table[Abs[eθ[θ]] (scale[-1, 1].rot[90 °]).{Cos[θ], Sin[θ]}, {θ, -π, π, (2 π)/60}];
elemmax = Max[(Sqrt[#1〚1〛^2 + #1〚2〛^2] &) /@ eθlist];
eθlist = eθlist/elemmax;  (* Normalize *)
eθgraphics = {RGBColor[1, 0, 0], Line[eθlist]};
ParametricPlot[Abs[Sin[θ]] (scale[-1, 1].rot[90 °]).{Cos[θ], Sin[θ]}, {θ, -π, π},
 PlotRange → {{-1, 1}, {-1, 1}},
 AxesStyle → RGBColor[0.01, 0.01, 0.01],
 PlotStyle → {RGBColor[0.01, 0.01, 0.01], Dashing[{0.02`, 0.02`}]},
 AspectRatio → Automatic,
 Epilog → eθgraphics]
```

```
Directivity = 1.65104 = 2.17757 dBi

Gain (inc. reflection) = 1.37616 = 1.3867 dBi
```

Out[45]=



EMF
C.A. Balanis, Antenna Theory, p.410.

```
η = 120. * π;
k = 2 * π;
l = 0.5;
Clear[a];
sol = Solve[10 == 2 * Log[l/a], a][[1]];
a = a /. sol; (* アンテナ半径 *)
Rr = η/(2 * π) * (EulerGamma + Log[k * l] - CosIntegral[k * l] +
      1/2 * Sin[k * l] * (SinIntegral[2 * k * l] - 2 * SinIntegral[k * l]) + 1/2 * Cos[k * l] *
       (EulerGamma + Log[k * l / 2] + CosIntegral[2 * k * l] - 2 * CosIntegral[k * l]));
Xm = η/(4 * π) * (2 * SinIntegral[k * l] + Cos[k * l] * (2 * SinIntegral[k * l] - SinIntegral[2 * k * l]) -
      Sin[k * l] * (2 * CosIntegral[k * l] - CosIntegral[2 * k * l] - CosIntegral[(2 * k * a^2)/l]));
Rin = Rr/Sin[k * l / 2]^2;
Xin = Xm/Sin[k * l / 2]^2;
Print["Rin=", Rin];
Print["Xin=", Xin];
```

```
Rin=73.1296

Xin=42.5445
```